ECE:3360 – Lab 2 Report

Oliver Emery

23 February 2022

1 Introduction

The goal of this lab was to construct a simple stopwatch using shift registers, 7-segment displays, and two buttons.

The stopwatch must implement two modes with different timer resolutions. In Mode I, the stopwatch should begin with "0.0" displayed. When the first button is pressed, the display should increment every 0.1 seconds, updating the display to "0.1", "0.2", and so on until the display reaches "9.9". Pushing the first button while the stopwatch is counting up should stop the counter and freeze the display. Pressing the first button while paused should continue the timer. Once 9.9 seconds have elapsed in Mode I, the display should flash "9.9" once every two seconds. Pressing the second button for less than a second in any state should stop the counter and reset the stopwatch to 0.

In Mode II the stopwatch should function identically to Mode I. However, the timer should increment in steps of 1 second instead of 0.1 seconds. Similarly, the stopwatch should now show "00" at the start and "99" for overflows.

Pressing the second button for more than 1 second should reset the stopwatch to 0 and alternate between modes I and II.

2 Schematic



Figure 1: schematic as implemented



Figure 2: physical implementation

3 Discussion

I used $1\,\text{k}\Omega$ resistors for the seven-segment display in order to keep the current through each segment well beneath the target value of $6\,\text{mA}$. I also included a standard $0.1\,\mu\text{F}$ decoupling capacitor to smooth ripples in the power supply.

3.1 Hardware Design

The final design was almost identical to that prescribed in the lab manual, with the exception of the four-digit display. Fortunately, the pinout of the 3461AS was nearly identical to that of the 5161AS, and only required one additional wire for each digit.

3.2 Buttons and Debouncing

Each button has an associated structure in SRAM containing relevant information. I aimed to keep the debouncing logic as simple as possible: register a change in button state if and only if it maintains that changed state over a minimum period of time. An unregistered change occurs when the button state in memory differs from the button's hardware state. The final implementation requires 50 stable samples at $1\,\text{ms}$ intervals to register a change.

Also included in the button structure is a duration field, incremented every $100\,\text{ms}$ when a button is in the pressed state. This value is used when button B is released to determine whether or not to change the mode.

3.3 Four-digit Seven-segment Display

In pursuit of extra credit, I created the design with a four-digit seven-sigment display. The 3461AS can only display one unique digit at a time, with power controlled by one pin for each digit. One study[1] found that humans can detect flicker at an extreme of $500\,\text{Hz}$. With a system clock of $16\,\text{MHz}$, a refresh rate as large as $1\,\text{KHz}$ still affords an entire 16,000 cycles to spend per refresh. For each refresh, the stopwatch must display each digit for a distinct but equal period of time.

The shift register supports clock speeds up to $20\,\text{MHz}$; the final implementation requires approximately $8\,\mu\text{s}$ (lines 574-634) to shift one byte into the shift register. No display digit is powered while the shift register is loaded. In order to maximize the duty cycle of each digit, it is ideal for each digit to be powered substantially longer than it is not during its 25% of each refresh. Lines 635-648 below produce a delay of approximately $48\,\mu\text{s}$, resulting in a final duty cycle of

$$25\% \cdot \frac{48}{8+48} \approx 21.4\%$$

for each digit. Unsurprisingly, reducing the duty cycle dims the digits. Each refresh in the final implementation takes approximately $230\,\mu\text{s}$ (lines 576-611), which corresponds to a refresh rate of $4.35\,\text{kHz}$. Indeed, no flicker is detectable with the naked eye.

To control power to the display digits, one I/O pin is assigned to each of the four digits. While this method is simple and convenient, it would also be feasible to use another shift register in lieu of I/O pins to control power to each digit. While this would increase I/O pin availability, it would increase hardware cost and increase code complexity marginally.

[1] https://www.nature.com/articles/srep07861

See source comments in Appendix A for more local and detailed discussion.

A Source Code Listing

```
;; project: ece3360-lab02
;; file:    main.S
;; date:    20220223
;; author:  Oliver Emery
;;
;;   The main subroutine of our program performs exactly one function: digit
;;   display. Because the 4x7 segment display I use requires multiplexing,
;;   we must continuously cycle power through each digit of the display, and at
;;   a high enough frequency to avoid flickering.
;;
;;   At 1ms intervals, a timer interrupt is called. If 100ms have passed, the
;;   stopwatch value is incremented. This happens first so the value will be
;;   incremented at strictly constant 100ms intervals. Next, both buttons are
;;   run through the debouncing algorithm, which calls the associated handler
;;   if a button state change is registered.
;;
;;   If we were to try handling all functionality in the main subroutine loop,
;;   cycles required would vary across iterations because of branching. While
;;   performing value increments at precise 100ms / 1.6M cycle intervals would
;;   still be possible, it would be unnecessarily complicated. Using timer
;;   interrupts allows us to separate the time-depedendent code from the rest.
;;
;;   Instead of using an entire IO pin for each digit of the display, I could
;;   have used none. I would have used an additional shift register chained to
;;   the current one. The bits of the new register would be used to control the
;;   currently powered digit.
;;
;;   There weren't enough subroutines or data being passed around, so I didn't
;;   try to establish any sort of calling convention. In anything called by an
;;   interrupt registers MUST be preserved to avoid nasty bugs, but I spread
;;   registers out across several of the display subroutines called in main()
;;   to avoid unneeded stack access.
;;
;; D0=DS40002061B (ATmega48A/PA/88A/PA/168A/PA/328/P Datasheet)
;;
.include "m328Pdef.inc"

;; ********************************* Defines *********************************
        ; inputs from pushbuttons
        .equ    P_BTN_A = PINB3
        .equ    P_BTN_B = PINB4

        ; outputs to shift register
        .equ    P_SER   = PINB0
        .equ    P_RCLK  = PINB1
```

```
46          .equ    P_SRCLK = PINB2
47
48          ; outputs to 4x7 segment digit pins
49          .equ    P_DIG0  = PIND4
50          .equ    P_DIG1  = PIND5
51          .equ    P_DIG2  = PIND6
52          .equ    P_DIG3  = PIND7
53
54          ; timer states
55          .equ    S_RESET = 0x01
56          .equ    S_COUNT = 0x02
57          .equ    S_STOP  = 0x04
58          .equ    S_OFLOW = 0x08
59
60          .equ    SEG_COUNT = 4
61          .equ    DIG_COUNT = 2
62
63          ; short blink on/off 2^(BLINK_POW-1) times every BLINK_LONG
64          .equ    BLINK_POW       = 2
65          ; in tenths of a second
66          .equ    BLINK_LONG      = 20
67          .equ    BLINK_SHORT     = 3
68
69          .equ    OFLOW_STATE_INIT = BLINK_LONG
70
71          ; debounce window
72          .equ    BTN_WND_MSEC    = 50
73
74          ; 100 is 1:1 / realtime
75          .equ    SUBDIV_MS       = 100
76          ; scaling factor of mode `B'
77          .equ    RES_MODESCALE   = 10
78
79          ; struct btn_s {
80          .equ    btn_pressed     = 0x00  ; 1 if button is pressed, else 0
81          .equ    btn_mask        = 0x01  ; 1 << PIN#
82          .equ    btn_dwnd        = 0x02  ; detect window
83          .equ    btn_duration    = 0x03  ; duration pressed
84          .equ    btn_handler     = 0x04  ; change handler subroutine
85          ; }
86          .equ    sz_btn          = 6
87
88
89  ;; ***************************** Global Variables ******************************
90  .dseg
91  .org 0x0100
92          subdiv_scaler:  .byte 1
```

```
93          mode_scaler:    .byte 1
94          ; stopwatch
95          current_state:  .byte 1
96          oflow_state:    .byte 1
97          ; value displayed on stopwatch
98          current_value:  .byte SEG_COUNT
99          ; a struct btn_s for button A and one for B
100         button_a:       .byte 6
101         button_b:       .byte 6
102
103
104 ;; ************************* Interrupt Vector Table ***************************
105 .cseg
106 ; [D0:7.7,12.4]
107         .org 0x0000     jmp __reset
108         ; counter0 compare match A handler
109         .org OC0Aaddr   jmp __isr_oc0a
110
111
112 ;; ******************************** Constants *********************************
113 .org INT_VECTORS_SIZE
114
115 ; [0-9], '-', + null byte to keep arvasm2 from complaining
116 digit_bits: .db \
117         0b00111111, 0b00000110, 0b01011011, 0b01001111, 0b01100110, \
118         0b01101101, 0b01111101, 0b00000111, 0b01111111, 0b01101111, \
119         0b01000000, 0
120
121
122 ;; ************************** Interrupt Handlers ****************************
123
124 ;; void __reset()
125 ;;
126 ;;     Called at system reset. Performs initialization tasks, and then transfers
127 ;;     to main().
128 ;;
129 __reset:
130         ; stack pointer must be defined before calls can be made or interrupts
131         ; enabled [D0:7.5]
132         ldi     r16, high(RAMEND)
133         out     SPH, r16
134         ldi     r16, low(RAMEND)
135         out     SPL, r16
136         ; configure MCU functions before interrupts are enabled
137         call    init
138         ; enable interrupts
139         sei
```

```
140         ; transfer execution to main
141         jmp     main
142
143 ;; void __isr_oc0a()
144 ;;
145 ;;     Timer 0 compare match A handler. Called every 16,000 cycles / 1 ms.
146 ;;     Invokes handler subroutine at 100 ms intervals and processes raw button
147 ;;     input.
148 ;;
149 __isr_oc0a:
150         push    r16
151         in      r16, SREG
152         push    r16
153         push    YL
154
155         lds     r16, subdiv_scaler
156         dec     r16
157         sts     subdiv_scaler, r16
158         brne    __isr_oc0a_fi           ; if (--subdiv_scaler == 0) {
159
160         ldi     r16, SUBDIV_MS
161         sts     subdiv_scaler, r16      ;     subdiv_scaler = SUBDIV_MS
162         rcall   every_subdiv            ;     every_subdiv()
163 __isr_oc0a_fi:                          ; }
164
165         ldi     YL, low(button_a)
166         rcall   debounce                ; debounce(button_a)
167         ldi     YL, low(button_b)
168         rcall   debounce                ; debounce(button_b)
169
170         pop     YL
171         pop     r16
172         out     SREG, r16
173         pop     r16
174         reti
175
176 ;; *************************** Setup and Entrypoint ****************************
177
178 ;; void memclr(Y: void*, r16: len)
179 ;;
180 ;;     clear up to r16 bytes of SRAM at YH:YL
181 ;;
182 memclr:
183         push    r16
184         push    r17
185         push    YL
186
```

```
187         clr     r17
188 memclr_loop:
189         st      Y+, r17
190         dec     r16
191         brne    memclr_loop
192
193         pop     YL
194         pop     r17
195         pop     r16
196         ret
197
198 ;; void init()
199 ;;
200 ;;     Called before interrupts are enabled. Configure I/O, timer module, and
201 ;;     power settings.
202 ;;
203 init:
204         ; Configure Timer/Counter 0 to generate an interrupt every 1 ms. This
205         ; is done with a combination of:
206         ;     * /64 clock prescaling
207         ;     * clear timer on compare match (CTC) mode
208         ;
209         ; In CTC mode, the timer counts to the value held in OCR0A, generates
210         ; an interrupt, and then is automatically reset to 0. This allows for
211         ; an additional arbitrary scaling factor (up to 256) on top of any
212         ; prescaling.
213         ;
214         ; With /64 prescaling and CTC mode with a compare value of 250, an
215         ; interrupt is generated every
216         ;
217         ;       (16 000 000 Hz / 64 / 250)^-1 = (1000 Hz)^-1 = 1 ms
218         ;
219         ; Register Configuration Documentation
220         ;     OCR0A     [D0:15.9.4]           output compare register
221         ;     TIMSK0    [D0:15.5,15.9.6]      compare interrupt enable
222         ;     TCCR0A    [D0:15.7.2,15.9.1]    ctc mode
223         ;     TCCR0B    [D0:17.1,15.9.2]      /64 prescaling
224         ldi     r16, 249
225         out     OCR0A, r16              ; OCR0A = 249
226         ldi     r16, 1 << OCIE0A
227         sts     TIMSK0, r16             ; TIMSK0 = 1 << OCIE0A
228         ldi     r16, 1 << WGM01
229         out     TCCR0A, r16             ; TCCR0A = 1 << WGM01
230         ldi     r16, 1 << CS01 | 1 << CS00
231         out     TCCR0B, r16             ; TCCR0B = 1 << CS01 | 1 << CS00
232
233         ; IO setup
```

```
234         ldi     r16, 1 << P_SER | 1 << P_RCLK | 1 << P_SRCLK
235         out     DDRB, r16               ; DDRB = 1<<DDB2 | 1<<DDB1 | 1<<DDB0
236         ; inputs
237         ldi     r16, 1 << P_BTN_A | 1 << P_BTN_B
238         out     PORTB, r16              ; PORTB = 1 << PORTB4 | 1 << PORTB3
239
240         ldi     r16, 1 << P_DIG0 | 1 << P_DIG1 | 1 << P_DIG2 | 1 << P_DIG3
241         out     DDRD, r16               ; DDRD = 0xf0
242         out     PORTD, r16              ; PORTD = 0xf0
243
244         ; enable sleep instruction and configure for idle mode [D0:10.11.1]
245         ldi     r16, 1 << SE
246         out     SMCR, r16               ; SMCR = 1 << SE
247
248         ret
249
250 ;; void main()
251 ;;
252 ;;     Main program entrypoint.
253 ;;
254 main:
255         ; Kind of a hack but works for a program this small. Since our data
256         ; starts at offset 0x0100 and is shorter than 256 bytes, it will never
257         ; be necessary to modify YH for indirect data references. ZH must be
258         ; free for modification and use by the LPM and ICALL instructions.
259         ldi     YH, 0x01
260         ldi     ZH, high(digit_bits << 1)
261
262         ; Initialize global state variables
263         ldi     r16, S_RESET
264         sts     current_state, r16      ; current_state = S_RESET
265         ldi     r16, SUBDIV_MS
266         sts     subdiv_scaler, r16      ; subdiv_scaler = SUBDIV_MS
267         ldi     r16, 0
268         sts     mode_scaler, r16        ; mode_scaler = 0
269
270         ; Initialize stopwatch value to zero
271         ldi     r16, SEG_COUNT
272         ldi     YL, low(current_value)
273         rcall   memclr
274
275         ; Initialize button structures
276         ldi     r16, sz_btn
277         ; button_a = { .mask = 1 << PIN_BUTTON0, .handler = button_a_changed }
278         ldi     YL, low(button_a)
279         rcall   memclr
280         ldi     r17, 1 << P_BTN_A
```

```
281         ldi     r18, high(button_a_changed)
282         ldi     r19, low(button_a_changed)
283         std     Y+btn_mask, r17
284         std     Y+btn_handler, r18
285         std     Y+btn_handler+1, r19
286
287         ; button_b = { .mask = 1 << PIN_BUTTON1, .handler = button_b_changed }
288         ldi     YL, low(button_b)
289         rcall   memclr
290         ldi     r17, 1 << P_BTN_B
291         ldi     r18, high(button_b_changed)
292         ldi     r19, low(button_b_changed)
293         std     Y+btn_mask, r17
294         std     Y+btn_handler, r18
295         std     Y+btn_handler+1, r19
296
297 main_forever:                   ;     do {
298         lds     r16, current_state
299         cpi     r16, S_OFLOW
300         brne    main_forever_show       ; if (current_state == S_OFLOW) {
301         lds     r16, oflow_state        ;     if (oflow_state &
302         sbrs    r16, (8 - BLINK_POW)    ;           (1 << (8 - BLINK_POW))) {
303         rjmp    main_forever_show       ;         sleep();
304         sleep                           ;         continue;
305         rjmp    main_forever            ;     }
306 main_forever_show:                      ; }
307         rcall   show_digits             ; show_digits();
308         rjmp    main_forever    ;     } while (1);
309
310
311 ;; ************************** Control Subroutines ***************************
312
313 ;; void every_subdiv()
314 ;;
315 ;;     Called every 100 ms. Handles incrementing stopwatch value, display
316 ;;     blinking in overflow state, and tracking button press duration.
317 ;;
318 every_subdiv:
319         push    r16
320         push    r17
321         push    YL
322
323         lds     r16, current_state
324         cpi     r16, S_COUNT
325         brne    every_subdiv_elsif      ; if (current_STATE == S_COUNT) {
326
327         lds     r16, mode_scaler
```

```
328         tst     r16
329         breq    every_subdiv_count_inc  ;     if (mode_scaler) {
330
331         dec     r16
332         sts     mode_scaler, r16        ;         if (--mode_scaler > 0) {
333         brne    every_subdiv_fi         ;             goto every_subdiv_fi
334
335         ldi     r16, RES_MODESCALE      ;         }
336         sts     mode_scaler, r16        ;         mode_scaler = RES_MODESCALE
337
338 every_subdiv_count_inc:                 ;     }
339         rcall   inc_value               ;     inc_value()
340         rjmp    every_subdiv_fi         ; }
341
342 every_subdiv_elsif:
343         cpi     r16, S_OFLOW
344         brne    every_subdiv_fi         ; else if (current_state == S_OFLOW) {
345
346         lds     r16, oflow_state
347         mov     r17, r16
348         andi    r16, 0xff >> BLINK_POW  ;     byte bwnd = oflow_state & 0x3f;
349         andi    r17, (0xff << (8 - BLINK_POW)) & 0xff
350         dec     r16                     ;     byte bctr = oflow_state & 0xc0;
351         breq    every_subdiv_oflow_blink    ; if (--bctr > 0) {
352         or      r16, r17
353         sts     oflow_state, r16        ;         oflow_state = bwnd | bctr;
354         rjmp    every_subdiv_fi
355 every_subdiv_oflow_blink:               ;     } else {
356         ldi     r16, 1 << (8 - BLINK_POW)
357         add     r17, r16                ;         bctr += 1 << (8- BLINK_POW);
358         brne    every_subdiv_oflow_blink_blip   ; if (!bctr) {
359         ori     r17, BLINK_LONG         ;             bctr |= BLINK_LONG;
360         rjmp    every_subdiv_oflow_blink_fi
361 every_subdiv_oflow_blink_blip:          ;         } else {
362         ori     r17, BLINK_SHORT        ;             bctr |= BLINK_SHORT;
363 every_subdiv_oflow_blink_fi:            ;         }
364         sts     oflow_state, r17        ;         oflow_state = bctr;
365 every_subdiv_fi:                        ; } }
366
367         ; Update duration counter on pressed buttons
368         ldi     YL, low(button_a)
369         rcall   button_inc_duration     ; button_inc_duration(button_a)
370         ldi     YL, low(button_b)
371         rcall   button_inc_duration     ; button_inc_duration(button_b)
372
373 every_subdiv_ret:
374         pop     YL
```

```
375         pop     r17
376         pop     r16
377         ret
378
379 ;; void button_a_changed(Y: *button, r16: is_pressed)
380 ;;
381 ;;     Called when button A is detected as pressed or released. Controls state
382 ;;     transitions caused by button A.
383 ;;
384 button_a_changed:
385         tst     r16
386         breq    button_a_changed_ret            ; if (!is_pressed) return;
387
388         lds     r16, current_state              ; switch (current_state) {
389         cpi     r16, S_RESET
390         brne    button_a_changed_case_count     ; case S_RESET:
391         ldi     r16, S_COUNT                    ;     current_state = S_COUNT;
392         rjmp    button_a_changed_sto            ;     break;
393 button_a_changed_case_count:
394         cpi     r16, S_COUNT
395         brne    button_a_changed_case_stop      ; case S_COUNT:
396         ldi     r16, S_STOP                     ;     current_state = S_STOP;
397         rjmp    button_a_changed_sto            ;     break;
398 button_a_changed_case_stop:
399         cpi     r16, S_STOP
400         brne    button_a_changed_sto            ; case S_STOP:
401         ldi     r16, S_COUNT                    ;     current_state = S_COUNT;
402 button_a_changed_sto:
403         sts     current_state, r16              ; }
404
405 button_a_changed_ret:
406         ret
407
408 ;; void button_b_changed(Y: *button, r16: is_pressed)
409 ;;
410 ;;     Called when button B is detected as pressed or released. Controls state
411 ;;     transitions caused by button B; regardless of the current state,
412 ;;     releasing B will revert the current state back to the RESET state.
413 ;;
414 ;;     Also toggles stopwatch timescale / "mode" if button was held for at
415 ;;     least <9> tenths of a second.
416 ;;
417 button_b_changed:
418         push    r16
419         push    YL
420
421         tst     r16
```

```
422         brne    button_b_changed_ret    ; if (is_pressed) return;
423
424         ldd     r16, Y+btn_duration
425         cpi     r16, 9
426         brlo    button_b_changed_no_modeswitch
427                                         ; if (btn->duration >= 9) {
428         lds     r16, mode_scaler
429         tst     r16
430         brne    button_b_changed_to_mode1   ; if (!mode_scaler) {
431
432         ldi     r16, RES_MODESCALE      ;         mode_scaler = RES_MODESCALE;
433         sts     mode_scaler, r16        ;     }
434         rjmp    button_b_changed_no_modeswitch
435 button_b_changed_to_mode1:              ;     else {
436         clr     r16                     ;         mode_scaler = 0;
437         sts     mode_scaler, r16        ;     }
438 button_b_changed_no_modeswitch:         ; }
439
440         clr     r16
441         std     Y+btn_duration, r16     ; btn->duration = 0;
442         ldi     r16, S_RESET
443         sts     current_state, r16      ; current_state = S_RESET;
444
445         ldi     r16, SEG_COUNT
446         ldi     YL, low(current_value)
447         rcall   memclr                  ; current_value = "0000";
448
449 button_b_changed_ret:
450         pop     YL
451         pop     r16
452         ret
453
454
455 ;; *************************** Button Subroutines ****************************
456
457 ;; void button_inc_duration(YL: *button)
458 ;;
459 ;;     Increment the duration field of the passed button.
460 ;;
461 button_inc_duration:
462         ldd     r16, Y+btn_pressed
463         tst     r16
464         breq    button_inc_duration_ret ; if (btn->pressed) {
465
466         ldd     r16, Y+btn_duration
467         inc     r16                     ;     // prevent overflow
468         breq    button_inc_duration_ret ;     if (btn->duration + 1) {
```

```
469         std     Y+btn_duration, r16     ;         btn->duration++;
470 button_inc_duration_ret:                 ;     }
471         ret                             ; }
472
473 ;; void debounce(YL: *button)
474 ;;
475 ;;     Sample and process raw button input data to reliably detect and handle
476 ;;     button events. Big idea: register a change in button state if and only
477 ;;     if it holds the changed state steady for a specified window of time.
478 ;;
479 debounce:
480         push    r0
481         push    r1
482         push    r16
483         push    ZH
484         push    ZL
485
486         clr     r16
487         in      r0, PINB
488         ldd     r1, Y+btn_mask
489         and     r0, r1
490         brne    debounce_notpressed
491         inc     r16
492 debounce_notpressed:            ; byte pressed = (PINB & btn->mask) ? 0 : 1;
493
494         ldd     r0, Y+btn_pressed
495         cp      r16, r0
496         breq    debounce_coda           ; if (btn->pressed != pressed) {
497
498         ldd     r0, Y+btn_dwnd
499         dec     r0
500         std     Y+btn_dwnd, r0
501         brne    debounce_ret            ; if (--btn->dwnd) return;
502
503         std     Y+btn_pressed, r16      ;     btn->pressed = pressed;
504         ldd     ZH, Y+btn_handler
505         ldd     ZL, Y+btn_handler+1
506         ; lol totally unnecessary with only 2 buttons
507         icall                           ;     btn->handler();
508 debounce_coda:                          ; }
509         ldi     r16, BTN_WND_MSEC
510         std     Y+btn_dwnd, r16         ; btn->dwnd = WND_MSC;
511
512 debounce_ret:
513         pop     ZL
514         pop     ZH
515         pop     r16
```

```
516         pop     r1
517         pop     r0
518         ret
519
520
521 ;; *************************** Display Subroutines ****************************
522
523 ;; void inc_value()
524 ;;
525 ;;     Increment the current stopwatch value in memory. If it hits the maximum,
526 ;;     enter overflow state and leave the value maximized.
527 ;;
528 inc_value:
529         push    r16
530         push    YL
531
532         ldi     YL, low(current_value)  ; byte i = 0;
533
534 inc_value_loop:
535         ld      r16, Y
536         inc     r16
537         cpi     r16, 10
538         brne    inc_value_exit  ; while (current_value[i] + 1 == 10) {
539
540         clr     r16
541         st      Y+, r16         ;     current_value[i++] = 0;
542
543         cpi     YL, low(current_value + DIG_COUNT)
544         brne    inc_value_loop  ;     if (i == DIG_COUNT) {
545
546         ; FIXME - should use DIG_COUNT
547         ldi     r16, 9
548         ldi     YL, low(current_value)
549         st      Y+, r16
550         st      Y, r16
551
552         ldi     r16, S_OFLOW
553         sts     current_state, r16      ; current_state = S_OFLOW;
554         ldi     r16, OFLOW_STATE_INIT
555         sts     oflow_state, r16        ; oflow_state = OFLOW_STATE_INIT;
556
557         rjmp    inc_value_ret           ; return; }
558
559 inc_value_exit:                 ; }
560         st      Y, r16          ; current_value[i]++;
561
562 inc_value_ret:
```

```
563         pop     YL
564         pop     r16
565         ret
566
567 ;; void show_digits()
568 ;;
569 ;;     Display current stopwatch value on 4x7 segment display. Gimped version
570 ;;     for 2 digits so our friendly TA doesn't have to wait a minimum of 999.9
571 ;;     seconds to verify our overflow functionality. Writes hyphens on the
572 ;;     outer digits and the current value on the middle two.
573 ;;
574 show_digits:
575         push    r16
576         push    r17
577         push    r18
578
579         ; decimal point position
580         clr     r18
581         lds     r1, mode_scaler
582         tst     r1
583         breq    show_digits_not_ones
584         inc     r18
585 show_digits_not_ones:
586
587         ldi     YL, low(current_value)
588
589         ldi     r16, 1
590         ld      r17, Y+
591         rcall   write_digit
592
593         ldi     r16, 2
594         ld      r17, Y
595         dec     r18
596         rcall   write_digit
597
598         ; hyphens on digits 0 and 3
599         ldi     r17, 10
600         clr     r18
601         clr     r16
602         rcall   write_digit
603         ldi     r16, 3
604         rcall   write_digit
605
606         pop     r18
607         pop     r17
608         pop     r16
609         ret
```

```
610
611 ;; void write_digit(r16: index, r17: charn, r18: decimal)
612 write_digit:
613         ldi     ZL, low(digit_bits << 1)
614         add     ZL, r17
615         lpm     r19, Z
616
617         tst     r18
618         breq    write_digit_no_dp
619         ori     r19, 1 << 7
620 write_digit_no_dp:
621
622         rcall   put_sr_byte
623
624         mov     r20, r16
625         inc     r20
626         ldi     r21, ~(1 << 4)
627         ; can we get a barrel shifter up in here plx
628 write_digit_while:
629         lsr     r21
630         dec     r20
631         brne    write_digit_while
632
633         swap    r21
634         andi    r21, 0xf0
635         out     PORTD, r21
636
637         ; This just needs to be decently longer than the time it takes to load
638         ; the shift register. Each digit only gets 25% of total display time,
639         ; so we want to maximize the proportion of on time to off time. As it
640         ; stands, write_digit up to here takes ~8us (off), and the remainder
641         ; takes ~48us (on); each digit is on for ~85% of its period.
642         ldi     r19, 255
643 write_digit_delay:
644         dec     r19
645         brne    write_digit_delay
646
647         ldi     r19, 0xf0
648         out     PORTD, r19
649
650         ret
651
652 ;; void put_sr_byte(r19: byte)
653 ;;
654 ;;     Put a byte into the shift register.
655 ;;
656 put_sr_byte:
```

```
657         ldi     r20, 8
658 put_sr_byte_while:
659         rol     r19
660         brcs    put_sr_byte_while_hibit
661         cbi     PORTB, P_SER
662         rjmp    put_sr_byte_wend
663 put_sr_byte_while_hibit:
664         sbi     PORTB, P_SER
665 put_sr_byte_wend:
666         ; trigger SRCLK, shifting SER into the shift register. note that there
667         ; is no need for a delay: even if SBI/CBI only took 1 clock cycle, the
668         ; SN74HC595N supports up to 20 MHz while the UNO runs at only 16 MHz
669         sbi     PORTB, P_SRCLK
670         cbi     PORTB, P_SRCLK
671
672         dec     r20
673         brne    put_sr_byte_while
674
675         ; trigger RCLK to transfer shift register data to the storage register
676         sbi     PORTB, P_RCLK
677         cbi     PORTB, P_RCLK
678
679         ret
680
681 .exit
```